Maple-assisted proof of formula for A29683

Robert Israel

18 December 2017

There are $2^6 = 64$ possible configurations for a 2×3 sub-array. Consider the 64×64 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×3 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and each 1 in that row is king-move adjacent to 1,2 or 3 1's), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 & b_3 \\ b_4 & b_5 & b_6 \end{bmatrix}$$

as $b+1$ where $b_1 b_2 b_3 b_4 b_5 b_6$ is the binary representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t,M,i;
     s:= floor((a-1)/8);
     if s <> (b-1) mod 8 then return 0 fi;
     s:= convert(s+8,base,2);
     r:= convert(8+floor((b-1)/8),base,2);
     t:= convert(8+ ((a-1) mod 8),base,2);
     M:= Vector(3);
     if s[1] = 1 and s[2] = 1 then M[1]:= 1; M[2]:= 1 fi;
     if s[2]=1 and s[3]=1 then M[2]:= M[2]+1; M[3]:= 1 fi;
     for i from 1 to 3 do if s[i]=1 then
       M[i]:= M[i]+r[i]+t[i];
       if i > 1 then M[i]:= M[i]+r[i-1]+t[i-1] fi;
       if i < 3 then M[i]:= M[i]+r[i+1]+t[i+1] fi;
       if M[i]=0 or M[i] > 3 then return 0 fi;
     fi od;
     1
  end proc:
  T:= Matrix(64,64, q):
```

Thus $a(n) = u\,T^n\,v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](64):
  v:= Vector(64):
  for i from 0 to 7 do u[8*i+1]:= 1; v[i+1]:= 1;
  od:
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v, n = 1 .. 10);
```

4, 43, 210, 1162, 6959, 39608, 226599, 1305725, 7497482, 43051551 **(1)**

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

(2)

$$P := t \to t^{21} - 5\,t^{20} + 3\,t^{19} - 47\,t^{18} + 50\,t^{17} - 148\,t^{16} + 215\,t^{15} - 235\,t^{14} + 431\,t^{13} - 237\,t^{12} + 411\,t^{11} - 101\,t^{10} + 114\,t^{9} + 30\,t^{8} - 57\,t^{7} + 25\,t^{6} - 24\,t^{5} + 2\,t^{4} \quad \textbf{(2)}$$

This turns out to have degree 21. Thus we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=0}^{21} p_i\,a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 21, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 12, corresponding to a factor of P.

```
> factor(P(t));
```

$$t^4\,(t^2+1)\,(t^3-t^2+2\,t-1)\,(t^{12}-4\,t^{11}-4\,t^{10}-37\,t^9+15\,t^8-21\,t^7+79\,t^6+7\,t^5+92\,t^4+49\,t^3+19\,t^2+20\,t-2) \quad \textbf{(3)}$$

```
> Q:= unapply(t^12-4*t^11-4*t^10-37*t^9+15*t^8-21*t^7+79*t^6+7*
  t^5+92*t^4+49*t^3+19*t^2+20*t-2, t);
```

$$Q := t \to t^{12} - 4\,t^{11} - 4\,t^{10} - 37\,t^9 + 15\,t^8 - 21\,t^7 + 79\,t^6 + 7\,t^5 + 92\,t^4 + 49\,t^3 + 19\,t^2 + 20\,t - 2 \quad \textbf{(4)}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 9.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to (t^5 - t^4 + 3\,t^3 - 2\,t^2 + 2\,t - 1)\,t^4 \quad \textbf{(5)}$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all n. This will certainly satisfy the order-9 recurrence

$$\sum_{i=0}^{9} r_i\,b(i+n) = \sum_{i=0}^{9} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = \ldots = b(8) = 0$.

```
> seq( u . Q(T) . T^n . v, n = 0 .. 8);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0 \quad \textbf{(6)}$$